

NAG Library Routine Document

F07CDF (DGTTRF)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07CDF (DGTTRF) computes the *LU* factorization of a real n by n tridiagonal matrix A .

2 Specification

```
SUBROUTINE F07CDF (N, DL, D, DU, DU2, IPIV, INFO)
INTEGER N, IPIV(N), INFO
REAL (KIND=nag_wp) DL(*), D(*), DU(*), DU2(N-2)
```

The routine may be called by its LAPACK name *dgttrf*.

3 Description

F07CDF (DGTTRF) uses Gaussian elimination with partial pivoting and row interchanges to factorize the matrix A as

$$A = PLU,$$

where P is a permutation matrix, L is unit lower triangular with at most one nonzero subdiagonal element in each column, and U is an upper triangular band matrix, with two superdiagonals.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

5 Parameters

- | | | |
|----|--|---------------------|
| 1: | N – INTEGER | <i>Input</i> |
| | <i>On entry:</i> n , the order of the matrix A . | |
| | <i>Constraint:</i> $N \geq 0$. | |
| 2: | DL(*) – REAL (KIND=nag_wp) array | <i>Input/Output</i> |
| | Note: the dimension of the array DL must be at least $\max(1, N - 1)$. | |
| | <i>On entry:</i> must contain the $(n - 1)$ subdiagonal elements of the matrix A . | |
| | <i>On exit:</i> is overwritten by the $(n - 1)$ multipliers that define the matrix L of the <i>LU</i> factorization of A . | |
| 3: | D(*) – REAL (KIND=nag_wp) array | <i>Input/Output</i> |
| | Note: the dimension of the array D must be at least $\max(1, N)$. | |
| | <i>On entry:</i> must contain the n diagonal elements of the matrix A . | |
| | <i>On exit:</i> is overwritten by the n diagonal elements of the upper triangular matrix U from the <i>LU</i> factorization of A . | |

4:	$DU(*)$ – REAL (KIND=nag_wp) array	<i>Input/Output</i>
Note: the dimension of the array DU must be at least $\max(1, N - 1)$.		
<i>On entry:</i> must contain the $(n - 1)$ superdiagonal elements of the matrix A .		
<i>On exit:</i> is overwritten by the $(n - 1)$ elements of the first superdiagonal of U .		
5:	$DU2(N - 2)$ – REAL (KIND=nag_wp) array	<i>Output</i>
<i>On exit:</i> contains the $(n - 2)$ elements of the second superdiagonal of U .		
6:	$IPIV(N)$ – INTEGER array	<i>Output</i>
<i>On exit:</i> contains the n pivot indices that define the permutation matrix P . At the i th step, row i of the matrix was interchanged with row $IPIV(i)$. $IPIV(i)$ will always be either i or $(i + 1)$, $IPIV(i) = i$ indicating that a row interchange was not performed.		
7:	$INFO$ – INTEGER	<i>Output</i>
<i>On exit:</i> $INFO = 0$ unless the routine detects an error (see Section 6).		

6 Error Indicators and Warnings

$INFO < 0$

If $INFO = -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

$INFO > 0$

Element $\langle value \rangle$ of the diagonal is exactly zero. The factorization has been completed, but the factor U is exactly singular, and division by zero will occur if it is used to solve a system of equations.

7 Accuracy

The computed factorization satisfies an equation of the form

$$A + E = PLU,$$

where

$$\|E\|_{\infty} = O(\epsilon)\|A\|_{\infty}$$

and ϵ is the **machine precision**.

Following the use of this routine, F07CEF (DGTRRS) can be used to solve systems of equations $AX = B$ or $A^T X = B$, and F07CGF (DGTCON) can be used to estimate the condition number of A .

8 Parallelism and Performance

Not applicable.

9 Further Comments

The total number of floating-point operations required to factorize the matrix A is proportional to n .

The complex analogue of this routine is F07CRF (ZGTTRF).

10 Example

This example factorizes the tridiagonal matrix A given by

$$A = \begin{pmatrix} 3.0 & 2.1 & 0 & 0 & 0 \\ 3.4 & 2.3 & -1.0 & 0 & 0 \\ 0 & 3.6 & -5.0 & 1.9 & 0 \\ 0 & 0 & 7.0 & -0.9 & 8.0 \\ 0 & 0 & 0 & -6.0 & 7.1 \end{pmatrix}.$$

10.1 Program Text

```
Program f07cdfe

!     F07CDF Example Program Text

!     Mark 25 Release. NAG Copyright 2014.

!     .. Use Statements ..
Use nag_library, Only: dgttrf, nag_wp
!     .. Implicit None Statement ..
Implicit None
!     .. Parameters ..
Integer, Parameter :: nin = 5, nout = 6
!     .. Local Scalars ..
Integer :: info, n
!     .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: d(:), dl(:, du(:, du2(:))
Integer, Allocatable :: ipiv(:)
!     .. Executable Statements ..
Write (nout,*) 'F07CDF Example Program Results'
Write (nout,*)
!     Skip heading in data file
Read (nin,*)
Read (nin,*) n

Allocate (d(n),dl(n-1),du(n-1),du2(n-2),ipiv(n))

!     Read the tridiagonal matrix A from data file

Read (nin,*) du(1:n-1)
Read (nin,*) d(1:n)
Read (nin,*) dl(1:n-1)

!     Factorize the tridiagonal matrix A
!     The NAG name equivalent of dgttrf is f06cdf
Call dgttrf(n,dl,d,du,du2,ipiv,info)

If (info>0) Then
    Write (nout,99999) 'The (', info, ',', info, ')', &
        ' element of the factor U is zero'
End If

!     Print details of the factorization

Write (nout,*) 'Details of factorization'
Write (nout,*) 'Second super-diagonal of U'
Write (nout,99998) du2(1:n-2)
Write (nout,*) 'First super-diagonal of U'
Write (nout,99998) du(1:n-1)
Write (nout,*) 'Main diagonal of U'
Write (nout,99998) d(1:n)
Write (nout,*) 'Multipliers'
Write (nout,99998) dl(1:n-1)
Write (nout,*)
```

```

Write (nout,*) ' Vector of interchanges'
Write (nout,99997) ipiv(1:n)

99999 Format (1X,A,I3,A,I3,A,A)
99998 Format (1X,8F9.4)
99997 Format (1X,5I9)
End Program f07cdfe

```

10.2 Program Data

```

F07CDF Example Program Data
      5                                :Value of N
      2.1   -1.0   1.9    8.0
  3.0    2.3   -5.0   -0.9    7.1
  3.4    3.6    7.0   -6.0
                                         :End of matrix A

```

10.3 Program Results

F07CDF Example Program Results

Details of factorization

Second super-diagonal of U
-1.0000 1.9000 8.0000

First super-diagonal of U
2.3000 -5.0000 -0.9000 7.1000

Main diagonal of U
3.4000 3.6000 7.0000 -6.0000 -1.0154

Multipliers
0.8824 0.0196 0.1401 -0.0148

Vector of interchanges
2 3 4 5 5
